
Kurze Einf̈uhrung in dieKASH3-Syntax
Vorlesung Kodierungstheorie Sommersemester 2006

Im folgendem soll eine kurze Einführung in dieKASH3-Syntax gegeben werden. Dies geschieht an-
hand von Beispielen, an denen die Befehle und zu beachtende Dinge erl̈autert werden.

1 Erste Schritte in KASH3

KASH3ist ein Interpreter um sich derKANT V4-C-Bibliotheken zu bedienen.KANT V4ist ein auf
algebraische Zahlentheorie spezialisiertes Computeralgebra-System, was an der TU Berlin von der
KANT-Gruppe entwickelt wurde.
KASH3startet man, indem in die Shell der Befehlkash3 eingeben wird. Es erscheint dann ein Banner
und ein Eingabepromptkash . Verlassen kann manKASH3wieder mitkash> quit; . Nach dem
Aufruf von KASH3gelangt man in einen Interpreter, d.h. es können Befehle eingegeben und sofort
ausgef̈uhrt werden.KASH3-Programme werden mitRead("Dateiname"); eingelesen. Mit dem
Befehl LogTo("Dateiname"); wird alles, was nach diesem Befehl ausgegeben wird, genauso
in der Datei ”Dateiname” gespeichert. Mit Eingabe vonLogTo(); wird dieser Vorgang beendet.
Führt man speicherintensive Berechnungen inKASH3durch, so kann manKASH3mit kash3 -m n
aufrufen, wobein die Anzahl der Bytes ist, die für KASHreserviert werden sollen.

Die Syntax vonKASH3ist der von der Programmiersprache Pascal teilweise rechtähnlich. Eine erste
Besonderheit ist, daßKASH3sowohl imperative als auch funktionale Elemente enthält. Wir werden
dies sp̈ater an geeigneten Beispielen sehen.

Folgende Ẅorter oder Befehle sind inKASH3reserviert und k̈onnen nicht als Variablen benutzt wer-
den:

and , do , else , end , fi , for , function , if , in , local , mod, not , od , or , repeat , return ,
then , until , while , quit .

Ansonsten kann jedes Wort als Variable benutzt werden, wie z.B.

kash> text := "Hello world";
Hello world
oder auch
kash> a := 3;
3,

wobei:= der Zuweisungsoperator ist. Folgt auf einer Eingabe ein Semikolon, so versuchtKASH3den
Befehl auszuf̈uhren. Ein wichtiges Merkmal vonKASH3ist, daß man kein Typmanagement benötigt.
Um ein kleines Beispiel zu geben, wird erst der Begriff Listein KASHerklärt. Eine Liste inKASH3
ist eine endliche angeordnetes Tupel von beliebigen Elementen. Ein einfaches Beispiel ist

kash> L := [1..10000] ;
[1..10000] .

Hier istL eine Liste, die aus den ersten 10000 ganzen Zahlen besteht. Man kann durchL[k] auf das

k-te Element vonL zugreifen, sofern die ListeL auch soviel Element enthält:

kash> L[312];
312 .

Ein anderes Beispiel für eine Liste ist

K := [1, "HALLO", ZX, i-> i * 2, , [] ];
[ 1, "HALLO", Univariate Polynomial Ring in X over Integer Ri ng ,
function ( i ) ... end, , [] ] .

Hier haben wir unterschiedliche Objekte in der Liste: eine ganze Zahl, ein ASCII-String, ein Polynom-
ring,eine Funktion, eine Leerstelle in der ListeL und eine leere Liste. M̈ochte man den Polynomring
aus der Liste K lesen, so schreibt man

kash> PolyRing := K[3];
Univariate Polynomial Ring in x over Integer Ring

und KASH3hat dann den Polynomring in der ListeL unter der VariablePolyRing gespeichert.
Mit einer Definition einer Variablen ist also der Typ automatisch festgelegt. Listen sind jedoch keine
Mengen: So erḧalt man durch Eingabe von

kash> L := [3,3,3,4,4,5,6];
[ 3, 3, 3, 4, 4, 5, 6]

eine Liste mit Elementen, die mehrfach auftauchen.KASHkennt aber auch Mengen im mathemati-
schen Sinne. Der Befehl hierfür istSet . Für obiges Beispiel ergibt sich dann

kash> L := Set([3,3,3,4,4,5,6]);
[ 3, 4, 5, 6].

Möchte man wissen, ob das Element5 in der MengeL liegt, so gibt man

kash> 5 in L;

ein und erḧalt

TRUE.

Als nächstes wollen wir eine Funktion inKASHdefinieren. Das folgende Beispiel soll eine Liste von
Zahlen in eine Liste von Buchstaben eines vorgegebenen Alphabetes umwandeln. Die Buchstaben der
Worte kommen aus einem AlphabetA, das wir als Liste definieren:

kash> A := [’A’,’B’,’C’,’D’,’E’,’F’,’G’,’H’,’I’,’J’,’K ’,’L’,’M’,
’N’,’O’,’P’,’Q’,’R’,’S’,’T’,’U’,’V’,’W’,’X’,’Y’,’Z’ ];

"ABCDEFGHIJKLMNOPQRSTUVWXYZ"

Um zu sehen wie man Funktionen definiert soll folgendes Beispiel dienen:

NumbersToText := function(L,alphabet)
return(List(L, i-> alphabet[i+1]));

end;

Der Funktionsname istNumbersToText und als Parameter erhält die Funktion eine ListeL von
ganzen Zahlen und eine Alphabetalphabet . Als nächstes kommt eine sogenannte Listenfunktion.
Der BefehlList verlangt eine Liste und eine Funktion, hier alsoL und i -> alphabet[i+1] .
Die Funktion i -> alphabet[i+1] bildet eine Zahli auf dasi+1 -te Element im Alphabet
ab. Sinnvollerweise sollten also die Einträge der ListeL nur um eins verminderte Positionen von
Elementen des Alphabetsalphabet sein. Der BefehlList geht alle Elemente der ListeL durch,
bildet sie bez̈uglich der vorgegebenen Abbildung ab und gibt eine neue Liste zur̈uck. Diese besteht
dann aus den Bildern voni-> alphabet[i+1] . Dabei wird die Reihenfolge eingehalten, d.h. in
der zur̈uckgegebenen Liste ist dasi -te Element das Bild vomi -ten Element der ListeL. Schließlich
wird die so erhaltene Liste mitreturn zurückgegeben. Das bedeutet, wenn man den Aufruf und die
Zuordnung

kash> text := NumbersToText([7,0,11,11,14], A);

macht, so ordnetKASHder Variabletext das Ergebnis des Aufrufes der FunktionsNumbersToText
zu, n̈amlich den Wert

kash> text;
"HALLO"

KASHunterscheidet zwischen globalen und lokalen Variablen. Das n̈achste Beispiel soll deutlich ma-
chen, was eine lokale und was eine globale Variable ist:

Vigenere := function(block, key,alphabet)

local S,B,Cipher;

S := TextToNumbers(key,alphabet);
B := TextToNumbers(block, alphabet);
Cipher := NumbersToText(List([1..Length(S)],

i-> (S[i]+B[i]) mod Length(alphabet)), alphabet);
return(Cipher);

end;

Zunächst haben wir wieder einen FunktionsnamenVigenere . Die Funktion verlangt die Parameter
Block, key undalphabet . Als lokale Variablen werden nunS,B undCipher geẅahlt. Dies
macht man durch den Befehllocal deutlich, dem eine Aufz̈ahlund lokaler Variablen folgt. Das
bedeutet, daß diese Variablen ihre Gültigkeit nur innerhalb der FunktionVigenere haben. Gibt man
z.B. folgendes in die Shell ein:

kash> S;

so gibtKASH

Error, Variable: ’S’ must have a value

zurück. Wurde aber vorher

kash> S := 3;

eingegeben, so ẅurde man dann

kash> S;
3

erhalten. Global heißt eine Variable, wenn sie wie oben definiert wurde, also nicht innerhalb einer
Funktion. Innerhalb der FunktionVigenere besitzt die VariableS den innerhalb der Funktion zuge-
wiesenen Wert, also hierS := TextToNumbers(key,alphabet) . Beim Verlassen der Funk-
tion hatS wieder seinen alten Wert3. WürdeS nicht in der Aufz̈ahlung der lokalen Variablen auf-
tauchen, so hatS auch innerhalb der Funktion den selben Wert wie außerhalb. Wie man sieht k̈onnen
auch innerhalb einer Funktion wieder Funktionen aufgerufen werden. Allerdings m̈ussen diese vorher
definiert werden.

2 Funktional versus imperativ

Als nächstes werden wir einige weitere Operatoren auf Listen kennenlernen. Gleichzeitig wird immer
eine imperative Variante angegeben.

Der OperatorFiltered verlangt als Parameter eine ListeL und eine boolsche Funktion, z.B. die
Funktion i -> IsPrime(i) , wobei IsPrime(n) = false ist wennn keine Primzahl war,
ansonsten wirdtrue zurückgegeben. Filtered gibt eine neue Liste zurück, die aus allen Elementenk
der ListeL besteht, f̈ur dieIsPrime(k)=true gilt.

kash> Filtered([1..50], i-> IsPrime(i));
[ 2, 3, 5, 7, 11, 13, 17, 19, 23, 29, 31, 37, 41, 43, 47 ] .

Wir erhalten also eine Liste der Primzahlenn ≤50. Die imperative Variante ẅare hier

kash> for xx in [1.. 50] do
if IsPrime(xx) then

Print(xx," ");
fi;

od;
2 3 5 7 11 13 17 19 23 29 31 37 41 43 47

Möchte man nun wissen, ob in der Menge[2342..2444] sich irgendeine Primzahl befindet, so
gibt man folgendes ein:

kash> ForAny([2342..2444], i-> IsPrime(i));
true .

kash> xx :=2341; b := false;
2341
false
kash> repeat

xx := xx+1;
b := IsPrime(xx);
until(b or xx > 2443);

kash> b;
true

Wir wissen also jetzt, daß sich in der Liste[2342..2444] mindestens eine Primzahl befindet. Die
kleinste bekommen wir mit

kash> First([2342..2444], i-> IsPrime(i));
2347

oder

kash> repeat
xx := xx+1;
b := IsPrime(xx);
until(b or xx > 2444);

kash> xx;
2347

Als nächstes m̈ochten wir wissen, ob in der ListeK := [2,4,10,12,18,20,22] nur gerade
Zahlen vorkommen, also

kash> L := [2,4,10,12,18,20,22];
[ 2, 4, 10, 12, 18, 20, 22 ]
kash> ForAll([2,4,10,12,18,20,22], j-> j mod 2 =0);
true

kash> while xx <= Length(L) do
b := (L[xx] mod 2 =0);
xx := xx+1;

od;
kash> b;
true

Eine weitere wichtige Listenoperation istListSplit(L,n) , welche eine gegebene ListeL in Teil-
listen der L̈angen aufteilt. Ist nunn kein Teiler der Listenl̈angeLength(L) und giltLength(L)=qn
+ r mit |r| < n undq ∈ N, so gibtKASH qListen der L̈angen und eine Liste der L̈anger zurück:

kash> A1 := "1234567890ABCDEFGHIJKLMNOPQRSTUVWXYZ)(,.= !?ab
cdefghijklmnopqrstuvwxyz ";;

kash> ListSplit(A1,15);
[ "1234567890ABCDE", "FGHIJKLMNOPQRST", "UVWXYZ)(,.=!? ab",

"cdefghijklmnopq", "rstuvwxyz " ]

Hierbei unterdr̈uckt das Doppelsemikolon hinter der ersten Eingabe die Ausgabe vonA1. Ben̈otigt
man dasn-fache kartesische Produkt einer gegebenen TeilmengeA, so erḧalt man dies durch die
Eingabe vonCartesian(A,..,A); , wobeiA n-mal innerhalb der runden Klammern erscheint.
Bei zu großemA kann es jedoch sein, daßKASHzuviel Speicher belegt und abgebrochen wird.
Um dies zu vermeiden, benutzt man obigen BefehlListSplit . Wendet man diesen aufA an und
teilt dadurchA z.B. in TeillistenA1,A2,A3 , so brechnet man jeweils das kartesische Produkt von
Cartesian(Ai,Aj,Ak); (i,j,k ∈ {1, 2, 3}).

3 Chinesischer Restsatz und endliche K̈orper

Als nächstes wollen wir den chinesischen Restsatz anwenden. Deshalb soll hier kurz erl̈autert werden,
wie man inKASHdamit arbeiten kann.

Seien z.B. Zahlenz1 = 17, z2 = 25 undz3 = 133 vorgegeben. Der Chinesische Restsatz besagt nun,
daß f̈ur n = z1 · z2 · z3 die Tatsache

Z/nZ ∼= Z/z1Z × Z/z2Z × Z/z3Z (1)

gilt. Wenn man von ein Element in dem Produktring auf der rechten Seite von (1) das zugehörige
Element vonZ/nZ berechnen m̈ochte, kann man dies mit dem Hauptsatz der simultanen Kongruenzen
tun. Dies bedeutet also, wenn man z.B. für ein Element(6, 24, 99) des Produktringes einα ∈ Z/nZ

berechnen m̈ochte, so tut man das inKASHmit

kash> ChineseRemainder([17,25,133],[6,24,99]);
20049

d.h. also das gesuchte Element inZ/56525Z war α = 20049 mit α ≡ 6 mod 17, α ≡ 24 mod 25
undα ≡ 99 mod 133.

Im RSA-Verfahren ist ist ein sogenannter RSA-Moduln vorgegeben, wobein = pq das Produkt
zweier ungerader Primzahlenp undq ist. Ist z.B.p = 309811 undq = 4000037, so berechnet man
die Ordnung der multiplikativen Gruppe vonZ/nZ. Diese ist(p−1)(q−1) = t = 1239251153160. Ist
nun einb ∈ N mit ggT(b, t) = 1, so l̈aßt sich eina ∈ N berechnen mitxab ≡ x mod n ∀x ∈ Z/nZ.
Ist z.B.b = 32321 so berechnet mana in KASH mit

kash> IntXGcd(32321,1239251153160);
[ 1, [ 11349226241, -296 ] ]

und erḧalt damit Zahlena, c ∈ Z mit at + cb = 1. Hierbei ista = 11349226241 undc = −296, was
man durch

kash> 11349226241 * b -296 * t;
1

verifiziert. Oftmals ist es erforderlich mit endlichen Körpern zu rechnen. Ein Endlicher Körperk wird
in KASHz.B. durch

kash> k:= FF(2,8);
Finite field of size 2ˆ8

erzeugt, wie man ihn z.B. bei AES benötigt. Die schon bekannten KörperZ/pZ für eine vorgegebene
Primzahlp erzeugt man mitFF(p,1) . Um Elemente direkt angeben zu können, benutzt man

kash> k := FF(31847,2);
Finite field of size 31847ˆ2
kash> a := FFGenerator(k);
w.

Man kann dann alle318472 − 1 von Null verschiedenen Elemente durch Potenzen vona erzeugen,
d.h.k× = 〈a〉 = {ak | k = 1, . . . , p2 − 1}.

Für Polynome aus einem K̈orperk benutzt man stattIntXGcd die FunktionPolyXGcd . Ist z.B.
p = 31847, so seik := Z/pZ. Nun definiert man sich ein Polynom ink[T ] wie folgt:

kash> k := FF(p,1);
Finite field of size 31847
kash> kT := PolyAlg(k,"T");
Univariate Polynomial Ring in T over GF(31847)
kash> f := Poly(kT, [1,202,2,0,452333,1]);
Tˆ5 + 202 * Tˆ4 + 2 * Tˆ3 + 6475 * T + 1

und um zuüberpr̈ufen, aus welcher Polynomalgebraf nun kommt gibt man

kash> PolyAlg(f);
Univariate Polynomial Ring in T over GF(31847)

ein. Istg ein zweites Polynom

kash> g := Poly(kT, [1,203,45095090,1]);
Tˆ3 + 203 * Tˆ2 + 31585 * T + 1

so kann man Elementeα, β ∈ k[T ] berechnen mitαf + βg = 1, nämlich durch

kash> PolyXGcd(f,g);
[ 1, 27363 * Tˆ2 + 8628 * T + 4864,

4484 * Tˆ4 + 18735 * Tˆ3 + 27737 * Tˆ2 + 25276 * T + 26984 ]

wobeiα = 27363T 2 + 8628T + 4864 undβ = 4484T 4 + 18735T 3 + 27737T 2 + 25276T + 26984
ist. Dies verifiziert man durch

kash> (27363 * Tˆ2 + 8628 * T + 4864) * f +
(4484 * Tˆ4 + 18735 * Tˆ3 + 27737 * Tˆ2 + 25276 * T + 26984) * g;

1.

Der chinesische Restsatz für Polynome ausk[T ], wobeik ein Körper ist, angewandt auff undg wäre
hier z.B.

kash> h := ChineseRemainder([f,g],[T,Tˆ2+2 * T]);
30798 * Tˆ7 + 7150 * Tˆ6 + 19282 * Tˆ5 + 28693 * Tˆ4 + 9470 * Tˆ3 + 28606 * Tˆ2 +

21484 * T + 9167

was man wieder durch

kash> h mod f;
T

und

kash> h mod g;
Tˆ2 + 2 * T

verifiziert.

